finite T1 spaces are discrete

Johan G. F. Belinfante and Tiffany D. Goble
2004 February 7

```
In[1]:= << goedel54.07b; << tools.m

        :Package Title: goedel54.07b      2004 February 7 at 4:25 p.m.

        It is now:  2004 Feb 9 at 13:44

        Loading Simplification Rules

        TOOLS.M                      Revised 2004 January 3

        weightlimit = 40
```

summary

Any **T1** topology on a finite space is discrete.

any PointClosed family of finite sets is finite

Lemma: Finiteness is unaffected by adding or deleting a single element.

```
In[2]:= SubstTest[member, union[u, v], FINITE,
          {u → singleton[y], v → dif[x, singleton[y]]}] // Reverse

Out[2]= member[intersection[x, complement[singleton[y]]], FINITE] == member[x, FINITE]

In[3]:= member[intersection[x_, complement[singleton[y_]]], FINITE] := member[x, FINITE]
```

Lemma. A collection **t** of finite sets which holds the relative complement in **U[t]** of some singleton must be a finite collection.

```
In[4]:= SubstTest[implies, and[member[u, t], subclass[t, w]], member[u, w],
            {u -> dif[U[t], singleton[x]], w -> FINITE}]

Out[4]= or[member[t, FINITE], not[member[intersection[complement[singleton[x]], U[t]], t]],
          not[subclass[t, FINITE]]] == True

In[5]:= (% /. {x -> x_, t -> t_}) /. Equal -> SetDelayed
```

If **t** is a point−closed collection of finite sets, and if there is a point **x** in **U[t]**, then **t** is finite.

```
In[6]:= Map[not, SubstTest[and, implies[and[p1, p2], p4],
          implies[and[p3, p4], p5], not[implies[and[p1, p2, p3], p5]],
          {p1 → member[x, U[t]], p2 → member[t, PointClosed], p3 → subclass[t, FINITE],
           p4 → member[dif[U[t], singleton[x]], t], p5 → member[t, FINITE]}]]

Out[6]= or[member[t, FINITE], not[member[t, PointClosed]],
          not[member[x, U[t]]], not[subclass[t, FINITE]]] == True
```

```
In[7]:= (% /. {t -> t_}) /. Equal -> SetDelayed
```

There are only two cases where **U[t]** has no elements, namely **t = 0** and **t = singleton[0]**. Both of these are finite sets.

```
In[8]:= subclass[succ[singleton[0]], FINITE] // AssertTest

Out[8]= subclass[succ[singleton[0]], FINITE] == True

In[9]:= subclass[succ[singleton[0]], FINITE] := True
```

Note that:

```
In[10]:= equal[union[FINITE, succ[singleton[0]]], FINITE]

Out[10]= True
```

This justifies adding the following rewrtite rule:

```
In[11]:= union[FINITE, succ[singleton[0]]] := FINITE
```

The main result of this section is the following restatement of these facts in which both of the variables have been eliminated.

```
In[12]:= Map[equal[V, class[t, equal[V, #]]] &, SubstTest[class, x,
            or[member[t, z], not[member[t, y]], not[member[x, u]], not[subclass[t, z]]],
            {u -> U[t], y -> PointClosed, z -> FINITE}]] // Reverse

Out[12]= subclass[intersection[PointClosed, P[FINITE]], FINITE] == True

In[13]:= subclass[intersection[PointClosed, P[FINITE]], FINITE] := True
```

Reintroducing the variable **t** yields:

```
In[14]:= SubstTest[implies, and[member[t, u], subclass[u, v]],
          member[t, v], {u -> intersection[PointClosed, P[FINITE]], v -> FINITE}]

Out[14]= or[member[t, FINITE], not[member[t, PointClosed]], not[subclass[t, FINITE]]] == True

In[15]:= (% /. t → t_) /. Equal → SetDelayed
```

---

some lemmas

This section contains a collection of lemmas needed in the sequel.

Lemma 1: If **image[RC[U[t]], t]** holds the empty set and all singletons contained in **U[t]**, and is closed under binary unions, then it holds every finite subset of **U[t]**.

```
In[16]:= SubstTest[implies, and[member[0, x],
           subclass[image[SINGLETON, y], x], subclass[image[CUP, cart[x, x]], x]],
             subclass[intersection[FINITE, P[y]], x], {x -> image[RC[U[t]], t], y -> U[t]}]

Out[16]= or[not[member[U[t], t]], not[subclass[
             image[CUP, cart[image[RC[U[t]], t], image[RC[U[t]], t]]], image[RC[U[t]], t]]],
           not[subclass[image[SINGLETON, U[t]], image[RC[U[t]], t]]],
           subclass[intersection[FINITE, P[U[t]]], image[RC[U[t]], t]]] == True
```

```
In[17]:= (% /. t -> t_) /. Equal -> SetDelayed
```

Lemma 2. If **x** is contained in **y**, then their intersection is **x**.

```
In[18]:= or[not[subclass[x, y]],
           not[subclass[intersection[x, y], z]], subclass[x, z]] // NotNotTest

Out[18]= or[not[subclass[x, y]], not[subclass[intersection[x, y], z]], subclass[x, z]] == True

In[19]:= or[not[subclass[x_, y_]],
           not[subclass[intersection[x_, y_], z_]], subclass[x_, z_]] := True
```

Lemma 3. If **t** is a set, and if every subset of **U[t]** is the relative complement of some member of **t**, then **t** holds every subset of **U[t]**.

```
In[20]:= Map[implies[member[t, u], #] &,
          SubstTest[implies, subclass[u, v], subclass[image[w, u], image[w, v]],
            {u -> P[U[t]], v -> image[RC[U[t]], t], w -> RC[U[t]]}]]

Out[20]= or[not[member[t, u]],
           not[subclass[P[U[t]], image[RC[U[t]], t]]], subclass[P[U[t]], t]] == True

In[21]:= or[not[member[t_, u_]],
           not[subclass[P[U[t_]], image[RC[U[t_]], t_]]], subclass[P[U[t_]], t_]] := True
```

Lemma 4. If **t** holds every subset of its sum class, then **t** is the power set of of its sum class.

```
In[22]:= or[equal[t, P[U[t]]], not[subclass[P[U[t]], t]]] // AssertTest

Out[22]= or[equal[t, P[U[t]]], not[subclass[P[U[t]], t]]] == True

In[23]:= (% /. t -> t_) /. Equal -> SetDelayed
```

the logic

The proof carried out in the next section is based on the following facts:

```
In[24]:= {implies[and[p2, p4], p3], implies[p1, p6], implies[p2, p7],
           implies[p5, p8], implies[p8, p9], implies[and[p6, p7, p9], p10],
             implies[and[p3, p4], p11], implies[and[p10, p11], p12],
           implies[and[p1, p12], p13], implies[p13, p14]} /.
         {p1 → member[t, binclosed[CAP]], p2 -> member[t, PointClosed], p3 -> member[t, FINITE],
           p4 -> member[t, P[FINITE]], p5 -> member[t, fix[UCLOSURE]],
             p6 -> member[image[RC[U[t]], t], binclosed[CUP]],
           p7 -> subclass[image[SINGLETON, U[t]], image[RC[U[t]], t]],
                 p8 -> member[U[t], t], p9 -> member[0, image[RC[U[t]], t]],
                 p10 -> subclass[intersection[FINITE, P[U[t]]], image[RC[U[t]], t]],
                 p11 -> equal[P[U[t]], intersection[FINITE, P[U[t]]]],
                 p12 -> subclass[P[U[t]], image[RC[U[t]], t]],
           p13 -> subclass[P[U[t]], t], p14 -> equal[P[U[t]], t]}

Out[24]= {True, True, True, True, True, True, True, True, True, True}
```

Since a direct proof takes too long, the argument will be broken up into two steps and the hypotheses will be bundled so that it takes less time for the **GOEDEL** program to check the validity of the reasoning.

bundled and split

In the first part, the four of the hypotheses are bundled together. It is shown that these four hypotheses imply that every finite subset of **U[t]** is the relative complement of a member of **t**.

```
In[25]:= Map[not, SubstTest[and, implies[p1, p6], implies[p1, p7],
           implies[p1, p8], implies[p8, p9], implies[and[p6, p7, p9], p10],
             not[implies[p1, p10]], {p1 → and[member[t, binclosed[CAP]],
              member[t, PointClosed], member[t, P[FINITE]], member[t, fix[UCLOSURE]]],
             p6 -> member[image[RC[U[t]], t], binclosed[CUP]],
            p7 -> subclass[image[SINGLETON, U[t]], image[RC[U[t]], t]],
                 p8 -> member[U[t], t], p9 -> member[0, image[RC[U[t]], t]],
                 p10 -> subclass[intersection[FINITE, P[U[t]]], image[RC[U[t]], t]]}]]

Out[25]= or[not[equal[t, Uclosure[t]]], not[member[t, PointClosed]],
           not[subclass[t, FINITE]], not[subclass[image[CAP, cart[t, t]], t]],
           subclass[intersection[FINITE, P[U[t]]], image[RC[U[t]], t]]] == True

In[26]:= (% /. t -> t_) /. Equal -> SetDelayed
```

The rest of the argument is given here. What is shown is that a point−closed topology in which every open set is finite must be a discrete topology.

```
In[27]:= Map[not, SubstTest[and, implies[p1, p10], implies[p1, p3],
             implies[and[p1, p3], p11], implies[and[p10, p11], p12],
           implies[and[p1, p12], p13], implies[p13, p14], not[implies[p1, p14]],
                     {p1 → and[member[t, binclosed[CAP]], member[t, PointClosed],
              member[t, P[FINITE]], member[t, fix[UCLOSURE]]], p3 → member[t, FINITE]],
                 p10 -> subclass[intersection[FINITE, P[U[t]]], image[RC[U[t]], t]],
                 p11 -> equal[P[U[t]], intersection[FINITE, P[U[t]]]],
                 p12 -> subclass[P[U[t]], image[RC[U[t]], t]],
            p13 -> subclass[P[U[t]], t], p14 -> equal[P[U[t]], t]}]]

Out[27]= or[equal[t, P[U[t]]], not[equal[t, Uclosure[t]]], not[member[t, PointClosed]],
           not[subclass[t, FINITE]], not[subclass[image[CAP, cart[t, t]], t]]] == True

In[28]:= (% /. t -> t_) /. Equal -> SetDelayed
```

The hypothesis that **t** satisfies the topology axioms is replaced with the statement that **t** belongs to the class **TOPS**.

```
In[29]:= Map[not, SubstTest[and, implies[p1, p2], implies[p1, p3],
           implies[and[p2, p3, p4, p5], p6], not[implies[and[p1, p4, p5], p6]],
           {p1 → member[t, TOPS], p2 -> equal[t, Uclosure[t]],
            p3 -> subclass[image[CAP, cart[t, t]], t], p4 → subclass[t, FINITE],
            p5 → member[t, PointClosed], p6 → equal[t, P[U[t]]]}]]

Out[29]= or[equal[t, P[U[t]]], not[member[t, PointClosed]],
           not[member[t, TOPS]], not[subclass[t, FINITE]]] == True

In[30]:= or[equal[t_, P[U[t_]]], not[member[t_, PointClosed]],
           not[member[t_, TOPS]], not[subclass[t_, FINITE]]] := True
```

Eliminating the variables produces a concise formulation of the theorem: any T1 topology on a finite sace is discrete.

```
In[31]:= Map[equal[V, #] &,
           SubstTest[class, t, implies[and[member[t, u], member[t, v]], equal[t, P[U[t]]]],
            {u → intersection[PointClosed, P[FINITE]], v → TOPS}]] // Reverse

Out[31]= subclass[intersection[T1, TOPS, P[FINITE]], range[POWER]] == True
```

```
In[32]:= subclass[intersection[T1, TOPS, P[FINITE]], range[POWER]] := True
```